Classification of Quadratic Forms

Quadratic Forms on $\mathbb{R}^2$

Eigenvalues: 0 (zero quadratic form)

Eigenvalues: 0, 1 (positive semi-definite quadratic form)

In this case $Q(\boldsymbol{x}) = (x_2)^2$

In[1]:=
```
Plot3D[y^2, {x, -3, 3}, {y, -3, 3}]
```

Out[1]=



(I) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = 1\}$ is the union of two parallel lines $x_2 = 1$ and $x_2 = -1$.

In[2]:= `Plot3D[{1, y^2}, {x, -3, 3}, {y, -3, 3}, Mesh → False, PlotStyle → {{}, {Opacity[0.75]}}]`

Out[2]=


In[3]:= `ContourPlot[y^2, {x, -3, 3}, {y, -3, 3}, Contours → {1}, ContourShading → False]`

Out[3]=


(II) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = 0\}$ is the x_1-axis

In[4]:=
```
ContourPlot[y^2, {x, -3, 3}, {y, -3, 3}, Contours → {0.00001}, ContourShading → False]
```

Out[4]=



(III) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = -1\}$ is empty.

Eigenvalues: 1, 1 (positive definite quadratic form)

In this case $Q(\boldsymbol{x}) = (x_1)^2 + (x_2)^2$

In[5]:=
```
Plot3D[x^2 + y^2, {x, -3, 3}, {y, -3, 3}, BoxRatios → {1, 1, 2}]
```

Out[5]=



(I) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = 1\}$ is the unit circle in $\mathbb{R}^2$: $(x_1)^2 + (x_2)^2 = 1$.

In[6]:=
```
Plot3D[{1, x^2 + y^2}, {x, -3, 3}, {y, -3, 3}, Mesh → False, PlotStyle → {{}, {Opacity[0.75]}},
 PlotRange → {0, 6}, BoxRatios → {1, 1, 3/2}, ClippingStyle → None]
```

Out[6]=



In[7]:=
```
ContourPlot[x^2 + y^2, {x, -3, 3}, {y, -3, 3}, Contours → {1}, ContourShading → False]
```

Out[7]=



(II) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = 0\}$ is the set which consists of only one point, the origin (0, 0).

In[8]:=
```
ContourPlot[x^2 + y^2, {x, -3, 3}, {y, -3, 3}, Contours → {0.0001}, ContourShading → False]
```

Out[8]=



(III) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = -1\}$ is the empty set.

Eigenvalues: 0, -1 (negative semi-definite quadratic form)

Eigenvalues: -1, -1 (negative definite quadratic form)

Eigenvalues: -1, 1 (indefinite quadratic form)

In this case $Q(\boldsymbol{x}) = -(x_1)^2 + (x_2)^2$

In[9]:=
```
Plot3D[-x^2 + y^2, {x, -3, 3}, {y, -3, 3}, BoxRatios → {1, 1, 2}]
```

Out[9]=



(I) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = 1\}$ is the hyperbola in $\mathbb{R}^2$: $-(x_1)^2 + (x_2)^2 = 1$.

In[10]:=
```
Plot3D[{1, -x^2 + y^2}, {x, -3, 3}, {y, -3, 3}, Mesh → False,
 PlotStyle → {{Opacity[0.5]}, {Opacity[0.7]}}, BoxRatios → {1, 1, 2}, ClippingStyle → None]
```

Out[10]=



In[11]:=
```
ContourPlot[-x^2 + y^2, {x, -3, 3}, {y, -3, 3}, Contours → {1}, ContourShading → False]
```

Out[11]=



(II) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = 0\}$ is the set which consists of two lines, $x_1 = x_2$ and $x_1 = -x_2$.

In[12]:=
```
Plot3D[{0, -x^2 + y^2}, {x, -3, 3}, {y, -3, 3}, Mesh → False,
 PlotStyle → {{Opacity[0.5]}, {Opacity[0.7]}}, BoxRatios → {1, 1, 2}, ClippingStyle → None]
```

Out[12]=



In[13]:=
```
ContourPlot[-x^2 + y^2, {x, -3, 3}, {y, -3, 3}, Contours → {0}, ContourShading → False]
```

Out[13]=



(III) The set $\{\boldsymbol{x} \in \mathbb{R}^2 : Q(\boldsymbol{x}) = -1\}$ is the hyperbola in $\mathbb{R}^2$: $(x_1)^2 - (x_2)^2 = 1$..

In[14]:=
```
Plot3D[{-1, -x^2 + y^2}, {x, -3, 3}, {y, -3, 3}, Mesh → False,
  PlotStyle → {{Opacity[0.5]}, {Opacity[0.7]}}, BoxRatios → {1, 1, 2}, ClippingStyle → None]
```

Out[14]=



In[15]:=
```
ContourPlot[-x^2 + y^2, {x, -3, 3}, {y, -3, 3}, Contours → {-1}, ContourShading → False]
```

Out[15]=



Quadratic Forms on $\mathbb{R}^3$

Eigenvalues: 0 (zero quadratic form)

In this case $Q(\boldsymbol{x}) = 0$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is empty.

(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is $\mathbb{R}^3$.

(III) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is empty.

Eigenvalues: 0, 0, 1 (positive semi-definite quadratic form)

In this case $Q(\boldsymbol{x}) = (x_3)^2$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is the union of two parallel planes $x_3 = 1$ and $x_3 = -1$ (x_1, x_2 are arbitrary).

In[16]:=
```
ContourPlot3D[z², {x, -5, 5}, {y, -5, 5}, {z, -5, 5}, Contours → {1}, Mesh → False,
 ContourStyle → {Opacity[0.5]}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]
```

Out[16]=



(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is the $x_1\, x_2$-plane.

In[17]:=
```
ContourPlot3D[z^2, {x, -5, 5}, {y, -5, 5}, {z, -5, 5}, Contours → {0.0001}, Mesh → False,
  ContourStyle → {Opacity[0.5]}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]
```

Out[17]=



(III) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is the empty set.

Eigenvalues: 0, 1, 1 (positive semi-definite quadratic form)

In this case $Q(\boldsymbol{x}) = (x_2)^2 + (x_3)^2$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is the circular cylinder parallel to x_1-axis whose directrix is the unit circle in $x_2\, x_3$-plane.

In[18]:=
```
ContourPlot3D[y^2 + z^2, {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours → {1}, Mesh → False,
 ContourStyle → {Opacity[0.5]}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]
```

Out[18]=



(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is the x_1-axis.

In[19]:=
```
ContourPlot3D[y^2 + z^2, {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours → {0.0001}, Mesh → False,
 ContourStyle → {Opacity[0.5]}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]
```

Out[19]=



(III) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is the empty set.

Eigenvalues: 1, 1, 1 (positive definite quadratic form)

In this case $Q(\boldsymbol{x}) = (x_1)^2 + (x_2)^2 + (x_3)^2$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is the unit sphere.

In[20]:=
```
ContourPlot3D[x^2 + y^2 + z^2, {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, Contours → {1}, Mesh → False,
  ContourStyle → {Opacity[0.5]}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]
```

Out[20]=



(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is the set which consists of only one point, the origin (0, 0, 0).

(III) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is the empty set.

Eigenvalues: 0, 0, -1 (negative semi-definite quadratic form)

In this case $Q(\boldsymbol{x}) = -(x_3)^2$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is the empty set.

(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is the $x_1 x_2$-plane.

(III) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is the union of two parallel planes $x_3 = 1$ and $x_3 = -1$ (x_1, x_2 are arbitrary).

Eigenvalues: 0, -1, -1 (negative semi-definite quadratic form)

In this case $Q(\boldsymbol{x}) = -(x_2)^2 - (x_3)^2$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is the empty set.

(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is the x_1-axis.

(III) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is the circular cylinder parallel to x_1-axis whose directrix is the unit circle in $x_2 x_3$-plane.

Eigenvalues: -1, -1, -1 (negative definite quadratic form)

In this case $Q(\boldsymbol{x}) = -(x_1)^2 - (x_2)^2 - (x_3)^2$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is the empty set.

(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is the empty set.

(III) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is the unit sphere.

Eigenvalues: -1, 1, 1 (indefinite quadratic form)

In this case $Q(\boldsymbol{x}) = -(x_1)^2 + (x_2)^2 + (x_3)^2$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is the one-sheet hyperboloid. This surface can be visualized as one branch of the hyperbola $-(x_1)^2 + (x_3)^2 = 1$ rotating about x_1-axis.

In[21]:=
```
ContourPlot3D[-x^2 + y^2 + z^2, {x, -5, 5}, {y, -5, 5}, {z, -5, 5},
 Contours → {1}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]
```

Out[21]=



In[22]:=
```
Manipulate[Show[ParametricPlot3D[{Sinh[s], Cosh[s] Sin[θ], Cosh[s] Cos[θ]},
    {s, -5, 5}, {θ, 0, tt}, PlotPoints → {50, 64}, Mesh → False],
  ParametricPlot3D[{Sinh[s], Cosh[s] Sin[tt], Cosh[s] Cos[tt]}, {s, -5, 5},
   PlotPoints → {50}, PlotStyle → {{Thickness[0.01], RGBColor[1, 0, 0]}}],
  PlotRange → {{-5, 5}, {-5, 5}, {-5, 5}}, AxesLabel → {x, y, z}], {tt, 0.01, 2 Pi}]
```

Out[22]=



(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is the circular cone whose axes is the x_1-axes.

(III) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is the two-sheet hyperboloid.

Eigenvalues: -1, -1, 1 (indefinite quadratic form)

In this case $Q(\boldsymbol{x}) = -(x_1)^2 - (x_2)^2 + (x_3)^2$

(I) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 1\}$ is the two-sheet hyperboloid. This surface can be visualized as the hyperbola $-(x_1)^2 + (x_3)^2 = 1$ rotating about x_3-axis.

In[23]:=
```
ContourPlot3D[-x^2 - y^2 + z^2, {x, -5, 5}, {y, -5, 5}, {z, -5, 5},
 Contours → {1}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]
```

Out[23]=



In[24]:=
```
Manipulate[Show[ParametricPlot3D[{Sinh[s] Cos[θ], Sinh[s] Sin[θ], Cosh[s]},
    {s, -5, 5}, {θ, 0, tt}, PlotPoints → {50, 64}, Mesh → False],
  ParametricPlot3D[{Sinh[s] Cos[θ], Sinh[s] Sin[θ], -Cosh[s]},
    {s, -5, 5}, {θ, 0, tt}, PlotPoints → {50, 64}, Mesh → False],
  ParametricPlot3D[{Sinh[s] Cos[tt], Sinh[s] Sin[tt], Cosh[s]}, {s, -5, 5},
   PlotPoints → {50}, PlotStyle → {{Thickness[0.01], RGBColor[1, 0, 0]}}],
  ParametricPlot3D[{Sinh[s] Cos[tt], Sinh[s] Sin[tt], -Cosh[s]}, {s, -5, 5},
   PlotPoints → {50}, PlotStyle → {{Thickness[0.01], RGBColor[1, 0, 0]}}],
  PlotRange → {{-5, 5}, {-5, 5}, {-5, 5}}, AxesLabel → {x, y, z}], {tt, 0.01, Pi}]
```

Out[24]=



(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = 0\}$ is the circular cone whose axes is the x_3-axes.

In[25]:=
```
ContourPlot3D[-x^2 - y^2 + z^2, {x, -5, 5}, {y, -5, 5}, {z, -5, 5},
 Contours → {0}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]
```

Out[25]=



(II) The set $\{\boldsymbol{x} \in \mathbb{R}^3 : Q(\boldsymbol{x}) = -1\}$ is the one-sheet hyperboloid. This surface can be visualized as one branch of the hyperbola $(x_1)^2 - (x_3)^2 = 1$ rotating about x_3-axis.

In[26]:= ContourPlot3D[$-x^2 - y^2 + z^2$, {x, -5, 5}, {y, -5, 5}, {z, -5, 5},
Contours → {-1}, PlotPoints → {100, 100, 100}, AxesLabel → {x, y, z}]

Out[26]=



In[27]:=
```
Manipulate[Show[ParametricPlot3D[{Cosh[s] Cos[θ], Cosh[s] Sin[θ], Sinh[s]},
    {s, -5, 5}, {θ, 0, tt}, PlotPoints → {50, 64}, Mesh → False],
  ParametricPlot3D[{Cosh[s] Cos[tt], Cosh[s] Sin[tt], Sinh[s]}, {s, -5, 5},
   PlotPoints → {50}, PlotStyle → {{Thickness[0.01], RGBColor[1, 0, 0]}}],
  PlotRange → {{-5, 5}, {-5, 5}, {-5, 5}}, AxesLabel → {x, y, z}], {tt, 0.01, 2 Pi}]
```

Out[27]=



Fun with Hyperboloid

In[28]:=
```
Options[Graphics3D]
```

Out[28]=
```
{AlignmentPoint → Center, AspectRatio → Automatic, AutomaticImageSize → False,
 Axes → False, AxesEdge → Automatic, AxesLabel → None, AxesOrigin → Automatic,
 AxesStyle → {}, Background → None, BaselinePosition → Automatic, BaseStyle → {},
 Boxed → True, BoxRatios → Automatic, BoxStyle → {}, ClipPlanes → None,
 ClipPlanesStyle → Automatic, ColorOutput → Automatic, ContentSelectable → Automatic,
 ControllerLinking → False, ControllerMethod → Automatic, ControllerPath → Automatic,
 CoordinatesToolOptions → Automatic, DisplayFunction :→ $DisplayFunction,
 Epilog → {}, FaceGrids → None, FaceGridsStyle → {}, FormatType :→ TraditionalForm,
 ImageMargins → 0., ImagePadding → All, ImageSize → Automatic, ImageSizeRaw → Automatic,
 LabelStyle → {}, Lighting → Automatic, Method → Automatic, PlotLabel → None,
 PlotRange → All, PlotRangePadding → Automatic, PlotRegion → Automatic,
 PreserveImageOptions → Automatic, Prolog → {}, RotationAction → Fit,
 SphericalRegion → Automatic, Ticks → Automatic, TicksStyle → {},
 TouchscreenAutoZoom → False, ViewAngle → Automatic, ViewCenter → Automatic,
 ViewMatrix → Automatic, ViewPoint → {1.3, -2.4, 2.}, ViewProjection → Automatic,
 ViewRange → All, ViewVector → Automatic, ViewVertical → {0, 0, 1}}
```

In[29]:=
```
Graphics3D[{{Thickness[0.01], RGBColor[0.7, 0.7, 0.7],
    Line[{Cos[#], Sin[#], 1} & /@ Range[0, 2 Pi, Pi/64]]}, {Thickness[0.01],
    RGBColor[0.7, 0.7, 0.7], Line[{Cos[#], Sin[#], -1} & /@ Range[0, 2 Pi, Pi/64]]},
   {Line[{{Cos[#], Sin[#], 1}, {Cos[#], Sin[#], -1}}] & /@ Range[0, 2 Pi, Pi/48]}},
  PlotRange → {{-1.1, 1.1}, {-1.1, 1.1}, {-1.1, 1.1}}]
```

Out[29]=



In[30]:=
```
Manipulate[Graphics3D[{{Thickness[0.007],
    RGBColor[0.7, 0.7, 0.7], Line[{Cos[#], Sin[#], 1} & /@ Range[0, 2 Pi, Pi/64]]},
   {Thickness[0.007], RGBColor[0.7, 0.7, 0.7],
    Line[{{0, 0, 1}, {Cos[θ + #], Sin[θ + #], 1}}] & /@ Range[0, 2 Pi, 2 Pi / 2]},
   {Thickness[0.007], RGBColor[0.7, 0.7, 0.7],
    Line[{Cos[#], Sin[#], -1} & /@ Range[0, 2 Pi, Pi/64]]}, {Thickness[0.005], Opacity[0.95],
    {Hue[# / (2 Pi)], Line[{{Cos[θ + #], Sin[θ + #], 1}, {Cos[#], Sin[#], -1}}]} & /@
     Range[0, 2 Pi, Pi/16]}},
  PlotRange → {{-1.1, 1.1}, {-1.1, 1.1}, {-1.1, 1.1}}], {θ, 0, 2 Pi}]
```

Out[30]=



In[31]:=
```
Show[Plot3D[3 x^2 - 2 y^2, {x, -3, 3}, {y, -3, 3}, Mesh → False, PlotStyle → {Opacity[0.6]}],
  Graphics3D[{Thickness[0.01], Blue, Line[Table[{Cos[t], Sin[t], 0}, {t, 0, 2 Pi, Pi/64}]]}],
  Graphics3D[{Thickness[0.01], Red,
    Line[Table[{Cos[t], Sin[t], 3 Cos[t]^2 - 2 Sin[t]^2}, {t, 0, 2 Pi, Pi/64}]]}],
  PlotRange → {{-2, 2}, {-2, 2}, {-3, 3}}, BoxRatios → {1, 1, 3/2}]
```

Out[31]=



In[32]:=
```
Show[Plot3D[1 x^2 + 6 x y + 1 y^2, {x, -3, 3},
   {y, -3, 3}, Mesh → False, PlotStyle → {Opacity[0.6]}],
  Graphics3D[{Thickness[0.01], Blue, Line[Table[{Cos[t], Sin[t], 0}, {t, 0, 2 Pi, Pi/64}]]}],
  Graphics3D[{Thickness[0.01], Red,
    Line[Table[{Cos[t], Sin[t], 1 Cos[t]^2 + 6 Cos[t] Sin[t] + 1 Sin[t]^2}, {t, 0, 2 Pi, Pi/64}]]}],
  PlotRange → {{-2, 2}, {-2, 2}, {-4, 4}}, BoxRatios → {1, 1, 4/2}]
```

Out[32]=



In[33]:=
```
Show[Plot3D[6 x^2 - 4 x y + 3 y^2, {x, -3, 3},
  {y, -3, 3}, Mesh → False, PlotStyle → {Opacity[0.6]}],
 Graphics3D[{Thickness[0.01], Blue, Line[Table[{Cos[t], Sin[t], 0}, {t, 0, 2 Pi, Pi/64}]]}],
 Graphics3D[{Thickness[0.01], Red,
   Line[Table[{Cos[t], Sin[t], 6 Cos[t]^2 - 4 Cos[t] Sin[t] + 3 Sin[t]^2}, {t, 0, 2 Pi, Pi/64}]]}],
 PlotRange → {{-2, 2}, {-2, 2}, {0, 8}}, BoxRatios → {1, 1, 4/2}]
```

Out[33]=